Métodos Numéricos para Equações Diferenciais

Fernando Deeke Sasse
Departamento de Matemática, UDESC - Joinville

2008/2

Soluções Exatas: Oscilações

Consideremos a equação que descreve um sistema forçado periodicamente, sem amortecimento:

```
> eq:=m*diff(x(t),t$2)+kappa*x(t)=sin(omega*t);
```

$$eq := m\left(\frac{d^2}{dt^2}\,x(t)\right) + \kappa\,x(t) = \sin(\omega\,t) \quad (1)$$

Solução geral da parte homogênea:

```
> eqh:=lhs(eq)=0;
```

$$eqh := m\left(\frac{d^2}{dt^2}\,x(t)\right) + \kappa\,x(t) = 0 \quad (2)$$

```
> solh:=dsolve(eqh,x(t));
```

$$solh := x(t) = \_C1\,\sin\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) + \_C2\,\cos\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) \quad (3)$$

```
> xh:=rhs(solh);
```

$$xh := \_C1\,\sin\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) + \_C2\,\cos\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) \quad (4)$$

Solução particular da eq. não homogênea:

```
> xp:=A*cos(omega*t)+B*sin(omega*t);
```

$$xp := A\cos(\omega\,t) + B\sin(\omega\,t) \quad (5)$$

```
> eq1:=subs(x(t)=xp,eq);
```

$$eq1 := m\left(\frac{\partial^2}{\partial t^2}\,(A\cos(\omega\,t) + B\sin(\omega\,t))\right) + \kappa\,(A\cos(\omega\,t) + B\sin(\omega\,t)) = \sin(\omega\,t) \quad (6)$$

```
> eq1;
```

$$m\left(-A\cos(\omega\,t)\,\omega^2 - B\sin(\omega\,t)\,\omega^2\right) + \kappa\,(A\cos(\omega\,t) + B\sin(\omega\,t)) = \sin(\omega\,t) \quad (7)$$

```
> ex1:=lhs(%)-rhs(%);
```

$$ex1 := m\left(-A\cos(\omega\,t)\,\omega^2 - B\sin(\omega\,t)\,\omega^2\right) + \kappa\,(A\cos(\omega\,t) + B\sin(\omega\,t)) - \sin(\omega\,t) \quad (8)$$

```
> E1:=coeff(ex1,cos(omega*t));
```

$$E1 := -m\,A\,\omega^2 + \kappa\,A \quad (9)$$

```
> E2:=coeff(ex1,sin(omega*t));
```

$$E2 := -m\,B\,\omega^2 + \kappa\,B - 1 \quad \textbf{(10)}$$

```
> solp:=solve({E1=0,E2=0},{A,B});
```

$$solp := \left\{A = 0,\, B = -\frac{1}{m\,\omega^2 - \kappa}\right\} \quad \textbf{(11)}$$

Aqui temos que supor que $\omega \neq \sqrt{\frac{\kappa}{m}}$

```
> assign(solp);
> xp;
```

$$-\frac{\sin(\omega\,t)}{m\,\omega^2 - \kappa} \quad \textbf{(12)}$$

Solução geral:

```
> xg:=xh+xp;
```

$$xg := \_C1\,\sin\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) + \_C2\,\cos\left(\frac{\sqrt{\kappa}\,t}{\sqrt{m}}\right) - \frac{\sin(\omega\,t)}{m\,\omega^2 - \kappa} \quad \textbf{(13)}$$

Exemplo 1 . Sejam

```
> kappa:=1; m:=1; omega:=2;
```

$$\kappa := 1$$

$$m := 1$$

$$\omega := 2 \quad \textbf{(14)}$$

e condições iniciais $x(0) = 1$, $x'(0) = -1$. Devemos então deteminar $\_C1$ e $\_C2$.

```
> E1:=subs(t=0,xg)=1;
```

$$E1 := \_C1\,\sin(0) + \_C2\,\cos(0) - \frac{1}{3}\,\sin(0) = 1 \quad \textbf{(15)}$$

```
> E1;
```

$$\_C2 = 1 \quad \textbf{(16)}$$

```
> E2:=subs(t=0,diff(xg,t))=-1;
```

$$E2 := \_C1\,\cos(0) - \_C2\,\sin(0) - \frac{2}{3}\,\cos(0) = -1 \quad \textbf{(17)}$$

```
> E2;
```

$$\_C1 - \frac{2}{3} = -1 \quad \textbf{(18)}$$

```
> sol2:=solve({E1,E2},{_C1,_C2});
```

$$sol2 := \left\{\_C1 = -\frac{1}{3},\, \_C2 = 1\right\} \quad \textbf{(19)}$$

```
> assign(sol2);
> xg;
```

$$-\frac{1}{3}\sin(t)+\cos(t)-\frac{1}{3}\sin(2t) \qquad (20)$$

```
> plot(xg,t=0..15);
```



Vamos encapsular os comandos anteriores para o oscilador forçado e amortecido.

```
> restart:
> m:=1.;g:=0.01;kappa:=1;w:=1;a:=-0.01;F:=sin(omega*t)*exp(a*t);
  x0:=1;v0:=1;
```

$$m := 1.$$
$$g := 0.01$$
$$\kappa := 1$$
$$w := 1$$
$$a := -0.01$$
$$F := \sin(\omega t)\, e^{-0.01\,t}$$
$$x0 := 1$$
$$v0 := 1 \qquad (21)$$

```
> eq:=m*diff(x(t),t$2)+g*diff(x(t),t)+kappa*x(t)=F:
> eqh:=lhs(eq)=0:
> solh:=dsolve(eqh,x(t)):
> xh:=rhs(solh):
```

```
> xp:=(A*cos(omega*t)+B*sin(omega*t))*exp(a*t):
> eq1:=subs(x(t)=xp,eq):
> eq1:
> ex1:=simplify(lhs(%)-rhs(%)):

> E1:=subs(omega=w,coeff(ex1,cos(omega*t))):

> E2:=subs(omega=w,coeff(ex1,sin(omega*t))):
> omega:=w:
> solp:=solve({E1=0,E2=0},{A,B}):
> assign(solp):
> xg:=xh+xp:
> e1:=subs(t=0,xg)=x0:
> e2:=subs(t=0,diff(xg,t))=v0:
> sol2:=solve({e1,e2},{_C1,_C2}):
> assign(sol2):
> xg:=xg;
```

$$xg := 1.505018813\, e^{-\frac{1}{200} t} \sin\left(\frac{1}{200} \sqrt{39999}\, t\right) - 99.\, e^{-\frac{1}{200} t} \cos\left(\frac{1}{200} \sqrt{39999}\, t\right) + 100.\, \cos(t)\, e^{-0.01 t} \quad (22)$$

```
> plot(xg,t=70..200, numpoints=700);
```